

---

Subject: Estimating Pregnancy Outcomes using Nepal DHS 2022

Posted by [PintuPaul](#) on Sun, 02 Jul 2023 07:28:27 GMT

[View Forum Message](#) <> [Reply to Message](#)

---

Dear all,

I want to replicate pregnancy outcomes: Table 5.14 of Nepal DHS 2022 report (p.127). I didn't find information on pregnancy outcomes (miscarriage, stillbirth, and abortion) in the calendar data. Can anyone help me construct these variables for NDHS 2022?

---

---

Subject: Re: Estimating Pregnancy Outcomes using Nepal DHS 2022

Posted by [Janet-DHS](#) on Thu, 13 Jul 2023 16:46:38 GMT

[View Forum Message](#) <> [Reply to Message](#)

---

Following is a response from DHS staff member, Tom Pullum:

We apologize for the delay in this response, due to travel. This was a DHS-8 survey and it included a pregnancy history. It's much easier to construct a table such as table 5.14 from a pregnancy history than from a calendar. The following Stata lines illustrate how to do this, using just one of the covariates in the table: place of residence.

```
use "...NPIR81FL.DTA" , clear
keep v001 v002 v003 v005 v025 p19* p32*
rename *_0* *_*
reshape long p19_ p32_, i(v001 v002 v003) j(order)
keep if p19<36
tab v025 p32 [iweight=v005/1000000], row
```

p32 is the outcome (birth, stillbirth, miscarriage, or abortion), and p19 is months since the outcome. The IR file is re-shaped into a file in which each of the outcomes is a case. The weight and covariates are carried along in the reshaping

---

---

Subject: Re: Estimating Pregnancy Outcomes using Nepal DHS 2022

Posted by [PintuPaul](#) on Sun, 23 Jul 2023 07:44:35 GMT

[View Forum Message](#) <> [Reply to Message](#)

---

Thank you for your assistance and advice regarding computation of the pregnancy outcomes in the last three years from the Nepal DHS 2022. I was able to process and match it with Table 5.14 in the survey report.

I have a related query and would be grateful for your assistance.

I would like to be able to create a variable in which I can identify:

1) The number of women who are either currently married (n=11,155) or living with a partner

(n=25) had a pregnancy ending in the past 3 years. The current marital status variable is v501.

2) Among these subgroups of women who had an abortion at any time during the same period (past three years).

In other words, I would like to figure out the number of women in addition to the pregnancy outcomes, so I can analyze the data in terms of women but not pregnancies or its outcomes. Hence, I would like to consider women as the unit of analysis

Thank you so much for your time and consideration.

---

Subject: Re: Estimating Pregnancy Outcomes using Nepal DHS 2022

Posted by [Janet-DHS](#) on Mon, 31 Jul 2023 16:57:04 GMT

[View Forum Message](#) <> [Reply to Message](#)

---

Following is a response from DHS staff member, Tom Pullum:

The following Stata program has women as units of analysis and includes the numbers of births, stillbirths, miscarriages, or abortions in the past three years. You can tabulate by current marital status, although this is not necessarily the marital status at the time of the pregnancy outcome.

```
use "...NPIR81FL.DTA", clear
```

```
keep v001 v002 v003 v005 v025 v501 p19* p32*
```

```
rename *_0* *_*
```

```
reshape long p19_ p32_, i(v001 v002 v003) j(order)
```

```
rename *_* *_*
```

```
*keep if p19<36
```

```
gen births=0
```

```
gen stillbirths=0
```

```
gen miscarriages=0
```

```
gen abortions=0
```

```
replace births=1 if p32==1 & p19<36
```

replace stillbirths=1 if p32==2 & p19<36

replace miscarriages=1 if p32==3 & p19<36

replace abortions=1 if p32==4 & p19<36

collapse (sum) births stillbirths miscarriages abortions (first) v005 v025 v501, by(v001 v002 v003)

label values v501 V501

tab v501 births [iweight=v005/1000000]

tab v501 stillbirths [iweight=v005/1000000]

tab v501 miscarriages [iweight=v005/1000000]

tab v501 abortions [iweight=v005/1000000]

tab1 births stillbirths miscarriages abortions if v501==1 | v501==2 [iweight=v005/1000000]

---